

Paper 243-27

SEM for Health, Business and Education

Diana D. Suhr, Ph.D.

University of Northern Colorado

Abstract

Structural Equation Modeling (SEM) is a comprehensive statistical approach to testing hypotheses about relations among observed and latent variables (measured variables and unmeasured constructs) (Hoyle, 1995). SEM takes a confirmatory rather than an exploratory approach, specifies intervariable relations a priori, and estimates measurement errors explicitly (Suhr, 1999). The purpose of this paper is to provide an introduction to the SEM statistical approach with examples from health, business, and education fields. SAS[®] code (PROC CALIS), diagrams, and results will be discussed. In the health field, a path analysis investigates the prediction of self-perceived illness with effects of exercise participation, self-perceived fitness, stressful life experiences, and hardiness for promoting stress resistance (Kline, 1998; Roth, Wiebe, Fillingim, & Shay, 1989). Relating to the business field, the example examines the relationship between academic success and career success (e.g., ACT score, cumulative grade point average, salary) (Schumacker & Lomax, 1996). The next example, in the education field, compares results from a baseline latent growth curve model (LGM) of reading achievement to results from an LGM including a categorical variable as a covariate. Examples range from beginner to advanced levels (e.g., path analysis/regression to LGM).

PROC CALIS

The PROC CALIS procedure (Covariance Analysis of Linear Structural Equations) estimates parameters and tests the appropriateness of structural equation models using covariance structural analysis. Although PROC CALIS was designed to specify linear relations, structural equation modeling (SEM) techniques have the flexibility to test nonlinear trends.

Structural Equation Modeling (SEM)

SEM is a methodology for representing, estimating, and testing a theoretical network of (mostly) linear relations between variables (Rigdon, 1998). A structural equation model is a hypothesized pattern of directional and nondirectional relationships among a set of observed (measured) and unobserved (latent) variables (MacCallum & Austin, 2000). A measured variable is directly measured whereas a latent variable is not directly or exactly measured. Two main goals in SEM are 1) to understand the patterns of correlation among a set of variables and 2) to explain as much of their variance as possible with the model specified (Kline, 1998).

All variables are classified as either dependent or independent. Any variable that has a one-way path (or arrow) pointing at it is classified as a dependent variable. All other variables are classified as independent variables. Relations between variables are of three types: 1) association e.g., correlation, covariance; 2) direct effect, directional relation between two variables, e.g., independent and dependent variables; and 3) indirect effect, effect of an independent variable on a dependent variable through one or more intervening or mediating variables (Hoyle, 1995).

Analysis of the relations among the variables to be analyzed takes place after formally stating a model (specification). A path diagram is a pictorial representation of a model. In the model, parameters are specified as fixed or free. Fixed parameters are given a fixed value (e.g., zero or one). Free parameters are estimated from the data. The criterion selected for parameter estimation is known as the discrepancy function. It provides a guideline to minimize the differences between the population covariance matrix, Σ , as estimated by the sample covariance, S ,

and the covariance matrix derived from the hypothesized model, $\Sigma(0)$.

Identification determines whether a single, unique value for each and every free parameter can be estimated from the observed data. Necessary conditions for model identification are 1) a scale must be established for every latent variable in the model, e.g., fix the variance of each latent variable to 1 or fix the value to 1 of one parameter associated with a latent variable directional influence; 2) the number of free parameters must not exceed the number of nonredundant elements in the sample covariance matrix or $q < p^*$ where q is the number of estimated parameters in the model, $p^* = p(p + 1)/2$, and p is the number of measured variables in the model. If the second condition is violated, the researcher has fewer data values than parameters to be estimated (MacCallum, 1995).

If a single, unique value cannot be obtained for one or more free parameters, the model is underidentified ($q > p^*$). If a value can be obtained through one and only one manipulation of the observed data, the model is just identified ($q = p^*$). If a value for one or more free parameters can be obtained in multiple ways from the observed data, the model is overidentified ($q < p^*$).

In the most common form of SEM, the purpose of the model is to account for variation and covariation of the measured variables (MVs). Path analysis (e.g., regression) tests models and relationships among MVs. Confirmatory factor analysis tests models of relationships between latent variables (LVs or common factors) and MVs which are indicators of common factors. Latent growth curve models (LGM) estimates initial level (intercept), rate of change (slope), structural slopes, and variances. Special cases of SEM are regression, canonical correlation, confirmatory factor analysis, and repeated measures analysis of variance (Kline, 1998).

The use of SEM could be impacted by

- the research hypothesis being testing
- the requirement of sufficient sample size (e.g., 5-20 cases per parameter estimate)
- measurement instruments
- multivariate normality
- parameter identification
- outliers
- missing data
- interpretation of model fit indices (Schumacker & Lomax, 1996).

A suggested approach to SEM analysis proceeds through the following process:

- review the relevant theory and research literature to support model specification
- specify a model (e.g., diagram, equations)
- determine model identification (e.g., if unique values can be found for parameter estimation; the number of degrees of freedom, df , for model testing is positive)
- select measures for the variables represented in the model and collect data
- conduct preliminary descriptive statistical analysis (e.g., scaling, missing data, collinearity issues, outlier detection)
- estimate parameters in the model
- assess model fit
- respecify the model if meaningful
- present and interpret the model.

Statistics

Traditional statistical methods normally utilize one statistical test to determine the significance of the analysis. Structural Equation modeling, however, relies on several statistical tests to determine the adequacy of model fit to the data. The chi-square test indicates the amount of difference between expected and observed covariance matrices. A chi-square value close to zero indicates little difference between the expected and observed covariance matrices. In addition, the probability level must be greater than 0.05 when chi-square is close to zero.

The Comparative Fit Index (CFI) is equal to the discrepancy function adjusted for sample size. CFI ranges from 0 to 1 with a larger value indicating better model fit. Acceptable model fit is indicated by a CFI value of 0.90 or greater (Hu & Bentler, 1999).

Root Mean Square Error of Approximation (RMSEA) is related to residual in the model. RMSEA values range from 0 to 1 with a smaller RMSEA value indicating better model fit. Acceptable model fit is indicated by an RMSEA value of 0.06 or less (Hu & Bentler, 1999).

If model fit is acceptable, the parameter estimates are examined. The ratio of each parameter estimate to its standard error is distributed as a z statistic and is significant at the 0.05 level if its value exceeds 1.96 and at the 0.01 level if its value exceeds 2.56 (Hoyle, 1995). Unstandardized parameter estimates retain scaling information of variables and can only be interpreted with reference to the scales of the variables. Standardized parameter estimates are transformations of unstandardized estimates that remove scaling and can be used for informal comparisons of parameters throughout the model. Standardized estimates correspond to effect-size estimates.

If unacceptable model fit is found, the model could be revised when the modifications are meaningful. Model modification involves adjusting a specified and estimated model by either freeing parameters that were fixed or fixing parameters that were free. The Lagrange multiplier test provides information about the amount of chi-square change that results if fixed parameters are freed. The Wald test provides information about the change in chi-square that results if free parameters are fixed (Hoyle, 1995).

Health Example

SEM techniques are easily applied to analyses in the health field. Application of SEM techniques have contributed to research on illness (Roth, Wiebe, Fillingim, & Shay, 1989), on exercise (Duncan, T. & McAuley, 1993; Duncan, T., Oman, & Duncan, S., 1994) and on substance use/abuse among adults (Curran, Harford, & Muthen, 1996) and adolescents (Curran, Stice, & Chassin, 1997; Duncan, T., Duncan, S., Alpert, Hops, Stoolmiller, & Muthen, 1997).

This example reanalyzes data from a study where researchers investigated the effects of hardiness, stress, fitness, and exercise on health problems (Roth, et al., 1989). College students (n=373) reported recent physical illness, recent stressful life events, current exercise participation levels, current perceived fitness levels, and hardiness components. Multiple regression and SEM analyses examined the effects related to illness.

Subjects

Subjects were 163 men and 210 women enrolled in an introductory psychology course at a southern United States university. The mean age of the subjects was 21.7 (sd = 5.5).

Assessments

Illness. Seriousness of Illness Rating Scale (Wyler, Masuda, & Holmes, 1968) is a self-report checklist of commonly recognized physical symptoms and diseases and provides a measure of current and recent physical health problems. Each item is associated with a severity level. A total illness score is obtained

by adding the severity ratings of endorsed items (symptoms experienced within the last month).

Stress. Life Experience Survey (Sarason, Johnson, & Segal, 1978) is a measure used to access the occurrence and impact of stressful life experiences. Subjects indicate which events have occurred within the last month and rate the degree of impact on a 7-point scale (-3 = extremely negative impact, 0 = no impact, 3 = extremely positive impact). In the study, the total negative event score was used as an index of negative life stress (the absolute value of the sum of negative items).

Fitness. Fitness Questionnaire (Roth & Fillingim, 1988) is a measure of self-perceived physical fitness. Respondents rate themselves on 12 items related to fitness and exercise capacity. The items are on an 11-point scale of 0 = very poor fitness to 5 = average fitness to 10 = excellent fitness. A total fitness score is calculated by summing the 12 ratings. Items include questions about strength, endurance, and general perceived fitness.

Exercise. Exercise Participation Questionnaire (Roth & Fillingim, 1988) assessed current exercise activities, frequency, duration, and intensity. An aerobic exercise participation score was calculated using responses to 15 common exercise activities and providing blank spaces to write in additional activities.

Hardiness. In the study, hardiness included components of commitment, challenge, and control. A composite hardiness score was obtained by summing Z scores from scales on each component. The challenge component included one scale whereas the other components included 2 scales. Therefore, the challenge Z score was doubled when calculating the hardiness composite score.

Commitment was assessed with the Alienation From Self and Alienation From Work scales of the Alienation Test (Maddi, Kobasa, & Hoover, 1979). Challenge was measured with the Security Scale of the California Life Goals Evaluation Schedule (Hahn, 1966). Control was assessed with the External Locus of Control Scale (Rotter, Seaman, & Liverant, 1962) and the Powerlessness Scale of the Alienation Test (Maddi, Kobasa, & Hoover, 1979).

Results

Tests were conducted to determine if variables as a whole predicted a significant proportion of the variance of the illness measure and whether each individual variance uniquely accounted for a significant proportion of that variance. A main effects regression model including stress, fitness, hardiness, exercise, and gender to predict illness accounted for approximately 20% of the variance. The SEM analysis excluding gender found hardiness mediated by stress and exercise mediated by fitness (Roth, et al., 1989).

Reanalysis

The data was reanalyzed with PROC REG (regression) and PROC CALIS (path analysis). Input data was in the form of a correlation matrix, means, and standard deviations. A correlation matrix standardizes values and loses the metric of the scales. A covariance matrix would have preserved the metric of each scale.

SAS Code

```
data illf1 (type=corr);
  input   _type_ $1-4   _name_ $6-13
         exercise 15-20   hardy 22-27
         fitness 29-34   stress 36-41
         illness 43-48;

cards;
n          373      373      373      373      373
mean      40.90    0.00    67.10   4.80    716.7
std       66.50    3.80    18.40   6.70    624.8
corr      exercise 1.00
corr      hardy   -0.03    1.00
corr      fitness 0.39    0.07    1.00
corr      stress -0.05   -0.23   -0.13    1.00
corr      illness -0.08   -0.16   -0.29    0.34    1.00
;;;
```




Figure 1. Regression/Path Analysis Model

```
linesq
  fitness = p3p1 exercise          + e3,
  stress  = p4p2 hardy             + e4,
  illness = p5p3 fitness + p5p4 stress + e5;
std
  exercise = vare1,
  hardy    = vare2,
  e3       = vare3,
  e4       = vare4,
  e5       = vare5;
var exercise fitness hardy stress illness;
```

Results

The regression analysis with backward elimination retained fitness and stress ($p < 0.0001$) in the model while removing exercise and then hardy. R-squared with all variables in the model is equal to 0.1835. R-squared for the revised model is equal to 0.177.

The path analysis model showed some interesting results in terms of model fit: chi-square = 0.000, $df = 0$, $p < 0.0001$ and CFI = 1.000 and RMSEA = 0.000. Similar to the regression analysis, significant parameter estimates were fitness ($z = -5.067$) and stress ($z = 6.000$) while parameter estimates for hardy ($z = -1.530$) and exercise ($z = 0.663$) were not significant.




Figure 2. Structural Equation Model - Illness

Table 1. Parameter Estimates – Path Analysis

Parameter	Variable	Estimate	Std. Error	z-value
p5p1	Exercise → illness	0.034	0.051	0.663
p5p2	Hardy → illness	-0.074	0.048	-1.530
p5p3	Fitness → illness	-0.260	0.051	-5.067
p5p4	Stress → illness	0.291	0.049	6.000
Variances				
e5	Illness - unexplained variance	0.817	0.060	13.64

Standardized and unstandardized parameter estimates were equal because input was in the form of a correlation matrix (standardized values, mean = 0, std = 1). The standardized equation is

$$illness = 0.034 * exercise - 0.260 * fitness - 0.074 * hardy + 0.291 * stress + 0.904 * e5$$

R-squared (1 - unexplained variance squared) from the path analysis model is equal to 0.1835 ($1 - 0.9036^2 = 1 - 0.8165$). The unexplained variance is the amount of variance that cannot be accounted for with the predictor variables.

The structural equation model with hardiness mediated by stress and exercise mediated by fitness showed acceptable fit on three measures, chi-square (11.078, $df = 5$, $p = 0.050$), CFI (0.961), and RMSEA (0.057).

Table 2. Parameter Estimates – Structural Equation Model

Parameter	Variable	Estimate	Std. Error	z-value
p3p1	exercise → fitness	0.390	0.048	8.169
P4p2	hardy → stress	-0.230	0.051	-4.559
P5p3	fitness → illness	-0.250	0.047	-5.316
p5p4	stress → illness	0.308	0.047	6.538
Variances				
varex	Exercise	1.000	0.073	13.650
varhr	Hardy	1.000	0.073	13.650
varft	Fitness	0.848	0.062	13.640
varst	Stress	0.947	0.069	13.640
e5	Illness - unexplained variance	0.823	0.603	13.640

SAS Code

Regression Model

Note: The PROC GLM procedure will not accept a correlation or covariance matrix as input. Therefore, the PROC REG procedure was run.

```
proc reg data=illf1;
  model illness = exercise hardy fitness stress
  /selection = backward;
```

Path Analysis Model

```
proc calis data=illf1 corr stderr;
  linesq
    illness = p5p3 fitness + p5p4 stress +
    p5p1 exercise + p5p2 hardy + e5;
  std
    exercise = varex,
    hardy    = varhr,
    fitness  = varft,
    stress   = varst,
    e5       = vare5;
  var exercise fitness hardy stress illness;
```

Structural Equation Model

```
proc calis data=illf1 corr stderr;
```

Unstandardized and standardized parameter estimates are equal due to input in the form of a correlation matrix (standardized, mean = 0, std = 1). The standardized equations are
 Fitness = 0.390*exercise + 0.921*e3
 Stress = -0.230*hardy + 0.973*e4
 Illness = 0.253*fitness + 0.311*stress + 0.917*e5

Discussion

The regression and SEM model specify different relationships between variables in the model. The models include the same predictor and predicted variables in different configurations. Although statistical tests of significance differ, the amount of variance explained in each model is equal.

Conclusion

The development of theoretical models prior to SEM data analysis is critical. The results of the SEM analysis can serve to support or refute previous research. The direction of parameter estimates indicates effects on illness. More hardness indicates less stress and less stress indicates less illness. More exercise indicates better fitness and less illness. The SEM analysis provides flexibility in determining the relationships between variables. Direct as well as indirect relationships between variables can be specified and estimated.

Business Example

SEM is a relatively new technique in the business field. The example shown appeared as an exercise in *A Beginner's Guide to Structural Equation Modeling* (Schumaker & Lomax, 1996). The model tests the relationship between academic and career (job) success. Academic success is measured with ACT score, cumulative grade point average (cgpa) and entry level test score. Career success is measured with entry level salary and promotion salary. Input data is in the form of a covariance matrix. Unlike the correlation matrix (standardized), the covariance matrix is in the metric of each measured variable. The model type is confirmatory factor analysis. The measured variables act, cgpa, and entry are designed to measure the latent variable academic success. The measured variables salary and promo measure the latent variable job success. The covariance between F1 and F2 determines the relationship between two latent variables, academic and job success.




Figure 3. Structural Equation Model – Business

SAS Code

```
data jobfl (type=cov);
  input _type_ $ _name_ $ act cgpa entry
        salary promo;
cards;
n      500      500      500      500      500
cov act      1.024
cov cgpa     0.792  1.077
cov entry   0.567  0.537  0.852
cov salary  0.445  0.424  0.518  0.670
cov promo   0.434  0.389  0.475  0.545  0.716
;;;
```

```
proc calis data=jobfl cov stderr;
  lineqs
    act      = 1*F1 + e1,
    cgpa     = p2f1*F1 + e2,
    entry    = p3f1*F1 + e3,
    salary   = 1*F2 + e4,
    promo    = p5f1*F2 + e5;
  std
    e1 = vare1,
    e2 = vare2,
    e3 = vare3,
    e4 = vare4,
    e5 = vare5,
    F1 = varF1,
    F2 = varF2;
  cov
    f1 f2 = covf1f2;
  var act cgpa entry salary promo;
```

Results

The model provided acceptable fit for CFI (0.929) but less than acceptable values for chi-square (111.279, df = 4, p < 0.0001) and RMSEA (0.232). All parameter estimates were significant at the 0.01 level (z > 2.56).

Table 3. Parameter Estimates – Business Model

Variable	Estimate	Standard Error	z-value
act←F1	1.000*	--	--
Cgpa←F1 (p2f1)	0.972	0.048	20.265
Entry←F1 (p3f1)	0.785	0.043	18.122
Salary←F2	1.000*	--	--
Promo←F2 (p5f1)	0.943	0.046	20.341
Variance			
e1	0.260	0.029	8.780
e2	0.355	0.033	10.820
e3	0.381	0.030	12.900
e4	0.092	0.023	4.050
e5	0.202	0.024	8.690
F1	0.764	0.067	11.350
F2	0.578	0.047	12.200
Covariance			
F1F2	0.485	0.042	11.470

*fixed value

Standardized equations	R-squared (1-error variance squared)
Act = 0.864*F1 + 0.504*e1	0.747 = (1 - 0.504 ²)
Cgpa = 0.819*F1 + 0.574*e2	0.670 = (1 - 0.574 ²)
Entry = 0.743*F1 + 0.669*e3	0.552 = (1 - 0.669 ²)
Salary = 0.929*F2 + 0.370*e4	0.863 = (1 - 0.370 ²)
Promo = 0.847*F2 + 0.532*e5	0.717 = (1 - 0.532 ²)
Covariance F1F2 = 0.729	

Discussion

Parameter estimates in this model are similar to factor loadings. Results show variables selected to measure latent constructs are good measures (high factor loadings). The hypothesis of a relationship between academic and job success is supported by the magnitude of the covariance, 0.729.

The model RMSEA (0.232), a measure of model residual, was much larger than the 0.06 critical value. The large value indicates variance unexplained by the variables in the model. The chi-square value of 111.279 indicates unacceptable model fit. A chi-square value not close to zero indicates a difference between the expected and observed covariance matrices.

Conclusion

The model provided insight into the relationship between career and academic success. However, a better model could have been specified with additional measured variables. A general “rule of thumb” is to have at least 3 measured variables for each latent construct. The job success latent variable should have been specified with at least 3 measured variables, e.g., add supervisor evaluation. Future investigation of academic and career success should include a thorough literature review of previous research.

Education Example

Few studies have applied Latent Growth Modeling (LGM) to investigate cognitive development or acquisition of academic skills. This example will contrast a baseline LGM model to an LGM model with gender as a covariate. The models show development of reading recognition achievement for 6- through 12-year old children.

Participants

Participants were part of the National Longitudinal Survey of Youth (NLSY79). The original NLSY79 sample design enabled researchers to study longitudinal experiences of different age groups as well as analyze experiences of women, Hispanics, Blacks, and economically disadvantaged. The NLSY79 is a nationally representative sample of 12,686 young men and women who were 14- to 22-years old when first surveyed in 1979 (Baker, Keck, Mott, & Quinlan, 1993).

As part of the NLSY79, mothers and their children have been surveyed biennially since 1986. Although the NLSY79 initially analyzed labor market behavior and experiences, the child assessments were designed to measure academic achievement as well as psychological behaviors. The child sample consisted of all children born to NLSY79 women respondents participating in the study since 1986.

The number of children interviewed for the NLSY79 from 1988 to 1994 ranged from 4,971 to 7,089. Due to attrition, the number of children who completed four assessments in reading recognition achievement was 1,188. A cohort-sequential design in a previous study permitted estimation of growth curves for 5- through 14-year old boys and girls (Suhr, 1999). For simplicity, the 6-year old cohort (both boys and girls) was used as the example in this paper (n = 352, 163 boys and 189 girls). Assessments were collected at ages 6-, 8-, 10-, and 12-years old for the 6-year old cohort.

Measurement Instrument

The PIAT (Peabody Individual Achievement Test) was developed following principles of item response theory (IRT). The PIAT Reading Recognition Subtest, provides an achievement measure of word recognition and pronunciation ability, essential components of reading (Dunn & Markwardt, 1970). Test-retest reliability of PIAT Reading Recognition ranged between 0.81 for kindergartners to a high of 0.94 for third graders (median = 0.89).

Models

The SEM models shown in Figures 4 and 5 specify latent variables, initial level (intercept) and rate of change (slope). A constant is regressed on the latent variables to determine the mean initial level and mean rate of change across the time period shown. The latent variables are regressed on the measured variables to estimate structural slopes and determine rate of change. The error terms specified indicate the amount of variance in each measured variable (reading achievement). The disturbance terms (error terms of latent variables) estimate the amount of unexplained variance for each latent variable. The correlation between the disturbance terms indicates the relationship between the unexplained variances.

A covariate (male) has been added to the baseline model. Figure 5 illustrates the covariate model. A constant is regressed on the covariate. The covariate is regressed on latent constructs, initial value and growth rate.

Both models were estimated and evaluated for model fit and significance of estimated parameters. Comparisons will be made to contrast the differences between the two models.




Figure 4. Latent Growth Model - Achievement

Note: The intercept is a constant equal to 1. Measured variables (reading recognition achievement) are rr6, rr8, rr10, and rr12 measured at ages 6, 8, 10, and 12 respectively. Variances of measured variables are estimated with variances of E6, E8, E10, and E12. Structural slopes are fixed at 0 and 1 for ages 6 and 8 respectively. Structural slopes are estimated by pv10f2 and pv12f2 for ages 10 and 12. Unmeasured variables (latent variables) are initial value (F1) and rate of change (F2). Variances of latent variables are denoted and estimated by D1 and D2. The relationship between latent variable variances is denoted and estimated with R. The intercept is regressed on the latent variables to estimate mean initial value (ml) and mean rate of change (ms).




Figure 5. Covariate Latent Growth Model

Note: A categorical covariate (male) is regressed on initial value (F1) and growth rate (F2).

SAS Code**Baseline Model**

```
proc calis data=coh6 ucov aug;
  lineqs
    rr6 = F1 + e6,
    rr8 = F1 + F2 + e8,
    rr10 = F1 + pv10f2 F2 + e10,
    rr12 = F1 + pv12f2 F2 + e12,
    F1 = ml intercept + D1,
    F2 = ms intercept + D2;

  std
    e6 = vare6,
    e8 = vare8,
    e10 = vare10,
    e12 = vare12,
    D1-D2 = varD1-varD2;
  cov
    D1 D2 = CD1D2;
  var rr6 rr8 rr10 rr12;
```

Model with Gender as a Covariate

The following LGM model includes the variable male as a covariate. The variable gender has been recoded into a variable "male" with the following code:

if gender eq 1 then male=1; else if gender eq 2 then male=0;

```
proc calis data = coh6 ucov aug;
  lineqs
    rr6 = F1 + e6,
    rr8 = F1 + F2 + e8,
    rr10 = F1 + pv10f2 F2 + e10,
    rr12 = F1 + pv12f2 F2 + e12,
    male = mm intercept + em,
    F1 = f1m male + ml intercept + D1,
    F2 = f2m male + ms intercept + D2;

  std
    e6 = vare6,
    e8 = vare8,
    e10 = vare10,
    e12 = vare12,
    em = varem,
    D1-D2 = varD1-varD2;
  cov
    D1 D2 = CD1D2;
  var rr6 rr8 rr10 rr12 male;
```

Results

The baseline model showed acceptable fit on CFI (0.992). Unacceptable fit was found with chi-square (40.235, df = 3, $p < 0.0001$) and RMSEA (0.188). The covariate model found acceptable fit for CFI (0.993) and unacceptable fit for chi-square (40.423, df = 5, $p < 0.0001$) and RMSEA (0.142). Values for fit indices changed a small amount with the addition of the gender covariate. However, differences between boys and girls in the parameter estimates could be investigated.

Discussion

Structural slopes in the baseline and covariate models were 0, 1, 1.609, and 2.082. Growth steps can be determined by finding differences between structural slopes. The growth steps were equal to 1, 0.609, and 0.473 (1-0, 1.609 - 1, and 2.082 - 1.609 respectively). Growth steps and structural slopes provide information about the shape of the developmental curve, e.g., nonlinear and a negatively accelerating function of time). Decomposition of effects to determine gender differences in structural slopes was not examined in this paper.

Differences between parameter estimates for boys and girls can be determined from the unstandardized equations of the covariate model. If male is equal to 1 (boys), initial value is equal to 19.544

(19.929 - 0.385) and growth rate is equal to 16.074 (17.762 - 1.688). For girls, male is equal to 0 and initial value is equal to 19.929, growth rate is equal to 17.762. Parameter estimates are shown in Table 5.

Table 4. Parameter Estimates – Baseline Model

	Estimate	SE	z-value
Mean Initial Value (F1)	19.750	0.319	62.012
Mean Growth Rate (F2)	16.980	0.392	43.376
Variance of Initial Value (D1)	23.181	3.640	6.370
Variance of Growth Rate (D2)	21.591	2.630	8.210
Covariance of Initial Value and Growth Rate(R)	9.462	2.389	3.960
Growth Scores			
Age 6	0.000*	--	--
Age 8	1.000*	--	--
Age 10 (pv10f2)	1.609	0.027	60.123
Age 12 (pv12f2)	2.082	0.039	53.578
Variances			
Age 6 (e6)	12.587	3.094	4.070
Age 8 (e8)	21.045	2.092	10.060
Age 10 (e10)	17.153	2.645	6.490
Age 12 (e12)	55.532	5.743	9.670

*fixed value

Standardized Equations for Baseline Model

```
rr6 = 0.985*F1 + 0.172*e6
rr8 = 0.537*F1 + 0.465*F2 + 0.121*e8
rr10 = 0.420*F1 + 0.585*F2 + 0.085*e10
rr12 = 0.357*F1 + 0.643*F2 + 0.131*e12
F1 = 0.972*V999 + 0.237*D1
F2 = 0.965*V999 + 0.264*D2
```

Table 5. Parameter Estimates – Covariate Model

	Estimate	SE	z-value
Mean Initial Value (F1)	19.929	0.432	46.186
Mean Growth Rate (F2)	17.762	0.479	37.093
Mean Male	0.463	0.027	17.399
Variance of Initial Value (D1)	23.183	3.636	6.380
Variance of Growth Rate (D2)	20.924	2.579	8.110
Covariance of Initial Value and Growth Rate(R)	9.264	2.372	3.900
Growth Scores			
Age 6	0.000*	--	--
Age 8	1.000*	--	--
Age 10 (pv10f2)	1.609	0.027	60.123
Age 12 (pv12f2)	2.082	0.039	53.578
Variances			
Age 6 (e6)	12.521	3.088	4.050
Age 8 (e8)	21.049	2.091	10.070
Age 10 (e10)	17.074	2.630	6.490
Age 12 (e12)	55.719	5.740	9.710
Male (em)	0.249	0.018	13.250
Covariate-Initial Value (F1 ← Male) (f1m)	-0.385	0.628	-0.612
Covariate-Growth Rate (F2 ← Male) (f2m)	-1.688	0.579	-2.915

*fixed value

Standardized equations for Covariate Model

```
rr6 = 0.985*F1 + 0.171*e6
rr8 = 0.537*F1 + 0.465*F2 + 0.121*e8
rr10 = 0.420*F1 + 0.585*F2 + 0.085*e10
rr12 = 0.357*F1 + 0.643*F2 + 0.131*e12
male = 0.681*V999 + 0.643*em
F1 = -0.013*male + 0.980*V999 + 0.237*D1
F2 = -0.065*male + 1.009*V999 + 0.260*D2
```

The covariate model adjusts baseline parameter estimates for initial value and growth rate. The change from the baseline to covariate model can be calculated to give a rough estimate of adjusted values. Initial value baseline parameter estimate plus the product of mean male (proportion male subjects) and parameter estimate for the regression of the covariate (male) on the latent construct (initial level) to calculate adjusted value, 19.750 +

$(0.464)(-0.385) = 19.750 - 0.178 = 19.572$. Values used in the calculation are shown in Tables 4 and 5. Note: adjusted value is slightly different due to rounding.

Follow the same procedure to determine adjusted growth rate. Growth Rate baseline parameter estimate plus the product of mean male (proportion) and parameter estimate for regression of covariate (male) on latent construct (growth rate) to find adjusted value, $16.980 + (0.463)(-1.688) = 16.980 - 0.782 = 16.198$.

Calculation of a rough estimate of adjusted values for girls can be calculated by using the opposite direction (sign) for the parameter estimates. Initial value baseline parameter estimate plus the product of proportion of girls and parameter estimate for the regression of the covariate (male) with reverse direction on the latent construct (initial level) to calculate adjusted value, $19.750 - (0.537)(0.385) = 19.750 + 0.207 = 19.957$. Values used in the calculation are shown in Tables 4 and 5. Note: adjusted value is slightly different due to rounding.

Follow the same procedure to determine adjusted growth rate. Growth Rate baseline parameter estimate minus the product of mean male (proportion) and parameter estimate for regression of covariate (male) on latent construct (growth rate) to find adjusted value, $16.980 - (0.463)(-1.688) = 16.980 + 0.960 = 17.886$.

Conclusion

LGM analysis illustrated, for this sample, that the magnitude of initial value and growth rate is greater for 6-year old girls than 6-year old boys in reading achievement. Including covariates in a SEM analysis allows for the investigation of group differences (e.g., gender) that may be overlooked in analysis of the group as a whole.

SEM Analysis

The purpose of this paper has been to provide an introduction, with examples, to SEM analysis in the fields of health, business, and education. "Basics" of SEM techniques have been explained and illustrated. However, the possibilities for model specification and estimation are far more complex. Advanced research designs could include a cohort sequential model to estimate development or change while handling planned missing data, latent class analysis to predict probability of group membership, and multi-level multi-trait models to examine relationships among groups and behaviors.

SEM analysis provides comprehensive statistical techniques to examine and investigate the relationships between measured variables and latent constructs. SEM is a priori: a model makes theoretical sense and its statistical correspondence to the data is reasonable. The basic statistic in SEM is covariance. Two main goals of SEM analysis are to understand patterns of covariance among a set of variables, and to explain as much of their variance as possible with the model specified by the researcher. SEM analysis is appropriate for experimental as well as nonexperimental research designs. Therefore, SEM analysis allows researchers to examine and investigate relationships between variables with flexible and comprehensive techniques.

References

- Baker, P. C., Keck, C. K., Mott, F. L. & Quinlan, S. V. (1993). NLSY79 child handbook. A guide to the 1986-1990 National Longitudinal Survey of Youth Child Data (rev. ed.) Columbus, OH: Center for Human Resource Research, Ohio State University.
- Curran, P. J., Harford, T. C., & Muthen, B. O. (1996). The relation between heavy alcohol use and bar patronage: A latent growth model. *Journal of Studies on Alcohol*, 57, 410-418.
- Curran, P. J., Stice, E., & Chassin, L. (1997). The relation between adolescent alcohol use and peer alcohol use: A longitudinal random coefficients model. *Journal of Consulting and Clinical Psychology*, 68(1), 130-140.

Duncan, T. E., Duncan, S. C., Alpert, A., Hops, H., Stoolmiller, M., & Muthen, B. (1997). Latent variable modeling of longitudinal and multilevel substance use data. *Multivariate Behavioral Research*, 32(3), 275-318.

Duncan, T. E., Duncan, S. C., Strycker, L. A., Li, F., & Alpert, A. (1999). An introduction to latent variable modeling: Concepts, issues, and applications. Mahwah, NJ: Lawrence Erlbaum Associates, Publishers.

Duncan, T. E., & McAuley, E. (1993). Social support and efficacy cognitions in exercise adherence: A latent growth curve analysis. *Journal of Behavioral Medicine*, 16(2), 199-218.

Duncan, T.E., Oman, R., & Duncan, S. C. (1994). Modeling incomplete data in exercise behavior research using structural equation methodology. *Journal of Sport and Exercise Psychology*, 16, 187-203.

Dunn, L. M. & Markwardt, J. C. (1970). Peabody individual achievement test manual. Circle Pines, MN: American Guidance Service.

Hahn, M. E. (1966). *California Life Goals Evaluation Schedule*. Palo Alto, CA: Western Psychological Services.

Hoyle, R. H. (1995). The structural equation modeling approach: Basic concepts and fundamental issues. In Structural equation modeling: Concepts, issues, and applications, R. H. Hoyle (editor). Thousand Oaks, CA: Sage Publications, Inc., pp. 1-15.

Hu, L. & Bentler, P. M. (1999). Cutoff criteria for fit indexes in covariance structure analysis: Conventional criteria versus new alternatives. *Structural Equation Modeling*, 6(1), 1-55.

Kline, R. B. (1998). *Principles and Practice of Structural Equation Modeling*. New York: The Guilford Press.

MacCallum, R. C. (1995). Model specification: Procedures, strategies, and related issues. In Structural equation modeling: Concepts, issues, and applications, R. H. Hoyle (editor). Thousand Oaks, CA: Sage Publications, Inc., pp. 16-29.

MacCallum, R. C. & Austin, J. T. (2000). Applications of structural equation modeling in psychological research. *Annual Review of Psychology*, 51, 201-226.

Maddi, S. R., Kobasa, S. C., & Hoover, M. (1979). An alienation test. *Journal of Humanistic Psychology*, 19, 73-76.

Rigdon, E. E. (1998). Structural equation modeling. In Modern methods for business research, G. A. Marcoulides (editor). Mahwah, NJ: Lawrence Erlbaum Associates, Publishers, pp. 251-294.

Roth, D. L., & Fillingim, R. B. (1988). *Assessment of self-reported exercise participation and self-perceived fitness levels*. Unpublished manuscript.

Roth, D. L., Wiebe, D. J., Fillingim, R. B., & Shay, K. A. (1989). "Life Events, Fitness, Hardiness, and Health: A Simultaneous Analysis of Proposed Stress-Resistance Effects". *Journal of Personality and Social Psychology*, 57(1), 136-142.

Rotter, J. B., Seeman, M., & Liverant, S. (1962). Internal vs. external locus of control: A major variable in behavior theory. In N. F. Washburne (Ed.), *Decisions, values, and groups* (pp. 473-516). Oxford, England: Pergamon Press.

Sarason, I. G., Johnson, J. H., & Siegel, J. M. (1978). Assessing the impact of life changes: Development of the Life Experiences Survey. *Journal of Consulting and Clinical Psychology*, 46, 932-946.

SAS[®] STAT, Version 6. Cary, N.C.: SAS Institute.

Schumacker, R. E. & Lomax, R. G. (1996). *A Beginner's Guide to Structural Equation Modeling*. Mahwah, New Jersey: Lawrence Erlbaum Associates, Publishers.

Suhr, D. D. (2000). An investigation of mathematics and reading achievement of 5- through 14-year olds using latent growth curve methodology. *Dissertation Abstracts International, A (Humanities and Social Sciences)*. Vol 60(12-A), 4323, US: Univ Microfilms International.

Wyler, A. R., Masuda, M., & Holmes, T. H. (1968). Seriousness of Illness Rating Scale. *Journal of Psychosomatic Research*, 11, 363-374.

About the Author

Diana Suhr is a Statistical Analyst in the Office of Institutional Research at the University of Northern Colorado. She holds a Ph.D. in Educational Psychology with an Emphasis in Research, Measurement, and Statistics. She has been a SAS programmer and Statistical Consultant since 1984. She has presented contributed and invited papers, been a Section Chair at Western Users of SAS Software Conferences (WUSS) and was WUSS 2001 Program Chair.

Contact Information

Diana D. Suhr, Ph.D.
University of Northern Colorado
Institutional Research
Greeley, CO 80639
Phone: 970-351-2193
Email: diana.suhr@unco.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration. Other brand and product names are trademarks or their respective companies.

Appendix – Covariate Model

Covariance Matrix

rr6	427.789					
rr8	759.083	1434.781				
rr10	971.088	1816.832	2349.655			
rr12	1135.707	2127.954	2731.191	3249.476		
male	9.145	16.313	21.094	24.892	0.464	
Intercept	19.886	36.564	47.239	55.533	0.464	1.003

Note: The intercept row indicates mean values for measured variables.

Means, Standard Deviations (from PROC MEANS procedure)

6-year old Cohort	Variable	Mean	Std
	rr6	19.830	5.784
	rr8	36.460	10.082
	rr10	47.105	11.156
	rr12	55.375	13.578

6-year old Boys	Variable	Mean	Std
	rr6	19.693	6.370
	rr8	35.129	10.853
	rr10	45.423	12.175
	rr12	53.601	14.537

6-year old Girls	Variable	Mean	Std
	rr6	19.947	5.242
	rr8	37.608	9.242
	rr10	48.556	10.002
	rr12	56.905	12.530